

# 密度汎関数理論計算コード Quantum Espresso のコード読みのメモ

## 1 はじめに

密度汎関数理論では、エネルギーを電子密度の汎関数として与える。変分原理に基づき、エネルギーの停留条件として得られるのが Kohn-Sham 方程式である。これは、自己無撞着 (SCF) 計算と呼ばれる方法によって解かれる。本稿では、DFT 計算のフリーソフトウェアである Quantum Espresso (QE) を対象とし、SCF 計算のコードを解説することを目指す。

なお、QE の最新版 (2024 年 1 月時点) は version 7.3 だが、本コード読みでは version 5.1 を扱う。パッケージは以下のリンクからダウンロードできる：

<https://gitlab.com/QEF/q-e/-/releases/qe-5.1.0>

SCF 計算の主なコードは `q-e-qe-5.1.0/PW/src/` に格納されている。

また、コード読みの際は以下の文献を参考にした：

- Giannozzi, Paolo, et al. "QUANTUM ESPRESSO: a modular and open-source software project for quantum simulations of materials." *Journal of physics: Condensed matter* 21.39 (2009): 395502.
- R.M. マーチン 著, 物質の電子状態 (下)

## 2 自己無撞着計算における主要なファイル




図 1: Kohn-Sham 方程式を解く手順

- `run_pwscf.f90`: scf 計算の起点
- `init_run.f90`: 初期の電子密度と有効ポテンシャルを決定
- `electrons.f90`: scf ループ。サブルーチン `electrons.scf` が scf サイクルを動かしている。
- `v_of_rho.f90`: 有効ポテンシャルを更新
- `c_bands.f90`: 与えられた Hamiltonian に対して対角化
- `sum_band.f90`: 電子密度を計算
- `mix_rho.f90`: in と out の電子密度を混合。次の iteration の入力の電子密度として使われる。(iteration 前後の電子密度の差分から、収束判定も行っている?)

### 3 平面波電子状態計算の理論

■ **ブロッホの定理** 結晶は、構成する格子の周期で同じ構造を繰り返しもった系である。ブロッホの定理は、同じ構造を繰り返しもった系において電子の波動関数がとるべき形を規定するものである。本節ではこの定理を群論に基づき示そう。

結晶内の格子点の座標を

$$\mathbf{R} = n_1 \mathbf{a}_1 + n_2 \mathbf{a}_2 + n_3 \mathbf{a}_3 \quad (1)$$

と表す。ただし  $n_i$  は任意の整数である。ある関数に作用させると座標を  $\mathbf{R}$  だけならず操作（並進操作）

$$\hat{T}_{\mathbf{R}} f(\mathbf{r}) = f(\mathbf{r} + \mathbf{R}) \quad (2)$$

を定義しよう。すると、

$$\hat{T}_{\mathbf{R}} \hat{T}_{\mathbf{R}'} f(\mathbf{r}) = f(\mathbf{r} + \mathbf{R} + \mathbf{R}') = \hat{T}_{\mathbf{R}'} \hat{T}_{\mathbf{R}} f(\mathbf{r}) \quad (3)$$

であるので、並進操作 (2) の集合はアーベル群をなす。アーベル群の既約表現は全て一次元である。<sup>1</sup> ゆえ

<sup>1</sup>アーベル群は、任意の群元  $G_i$  と  $G_j$  に対し

$$G_i G_j = G_j G_i \quad (4)$$

が常に成り立つような集合のことである。また、任意の群元  $G_i$  に対し、同じ群の元  $G$  によって  $G^{-1} G_i G = G_j$  のように変換した元  $G_j$  を、 $G_i$  に共役な元という。 $G_i$  に共役な元全ての集合を、 $G_i$  の類と呼ぶ。アーベル群では、(4) から明らかのように、どの元も単独で類をなす。ゆえに、類の個数は群の位数に等しい。

今、群  $G$  の既約表現を  $D^{(\alpha)}$ 、位数を  $g$  とする。群論における重要な定理の一つとして知られる大直交定理

$$\sum_G D_{ij}^{(\alpha)*}(G) D_{kl}^{(\beta)}(G) = \frac{g}{d_\alpha} \delta_{\alpha\beta} \delta_{ik} \delta_{jl} \quad (5)$$

から、次の指標  $\chi^{(\alpha)}(G)$  に関する直交性の定理を導出することができる：

$$\sum_G \chi^{(\alpha)*}(G) \chi^{(\beta)}(G) = g \delta_{\alpha\beta} \quad (6)$$

同じ類での元の指標は一致する。ゆえに、類  $C_k$  に含まれる元の数  $h_k$  とすると

$$\sum_{k=1}^{n_c} h_k \chi^{(\alpha)*}(C_k) \chi^{(\beta)}(C_k) = g \delta_{\alpha\beta} \quad (7)$$

が成り立つことがわかる。また、証明は省くがこの式から

$$\sum_{\alpha=1}^{n_r} \chi^{(\alpha)*}(C_i) \chi^{(\beta)}(C_j) = \delta_{ij} \frac{g}{h_i} \quad (8)$$

を導くことができる。これは指標に関する二番目の定理である。

(7) は、次のような幾何学的な解釈ができる。

$$(\sqrt{h_1} \chi_1^{(\alpha)}, \sqrt{h_2} \chi_2^{(\alpha)}, \dots, \sqrt{h_{n_c}} \chi_{n_c}^{(\alpha)}) \quad (9)$$

を  $n_c$  次元のベクトルと見做せば、(7) はベクトルの内積が直交していることを表す。 $n_c$  次元空間において直交しているベクトルは高々  $n_c$  個であるので、

$$n_r \leq n_c \quad (10)$$

である。同様に、(8) から

$$n_c \leq n_r \quad (11)$$

が成り立つ。よって、 $n_c = n_r$  が結論される。つまり、類の数と既約表現の数は等しい。

$C_k$  を単位元とみなすと、単位元は単独で類をなす（つまり  $h_k = 1$ ）。単位元の指標は既約表現の次元  $d_\alpha$  に一致することから、(7) より

$$\sum_{k=1}^{n_c} d_\alpha^2 = g \quad (12)$$

が得られる。アーベル群では、類の数と群の位数は一致するため、 $d_\alpha = 1$  とならなければならないことがわかる。

に、 $\hat{T}_{\mathbf{R}}$  を  $\psi(\mathbf{r})$  に作用させた結果は、任意の複素数  $C_{\mathbf{R}}$  を用いて

$$\hat{T}_{\mathbf{R}}\psi(\mathbf{r}) = C_{\mathbf{R}}\psi(\mathbf{r}) \quad (13)$$

と書ける。また、全ての  $\hat{T}_{\mathbf{R}}$  は  $\hat{T}_{\mathbf{a}_i}$  を基本単位 (生成元) として作ることができる。つまり、 $\hat{T}_{\mathbf{R}} = (\hat{T}_{\mathbf{a}_1})^{n_1}(\hat{T}_{\mathbf{a}_2})^{n_2}(\hat{T}_{\mathbf{a}_3})^{n_3}$  である。よって、 $\hat{T}_{\mathbf{R}}$  は巡回群でもある。

ここで、結晶の巨視的な境界条件を与える。今、結晶のサイズが  $N_1\mathbf{a}_1 \times N_2\mathbf{a}_2 \times N_3\mathbf{a}_3$  であるとする。ここで、 $N_i$  は巨視的な数字である。境界条件として、このサイズの結晶がさらに全方向に繰り返すものとする。これはボルン・フォン・カルマン条件と呼ばれる。この条件下では、結晶内に広がった電子の波動関数は

$$\psi(\mathbf{r} + N_i\mathbf{a}_i) = \psi(\mathbf{r}) \quad (14)$$

を満たすと考える。上式は、並進操作を用いて

$$\hat{T}_{N_i\mathbf{a}_i}\psi(\mathbf{r}) = \psi(\mathbf{r}) \quad (15)$$

と書き表すことができる。よって、独立な並進操作の数は  $N = N_1N_2N_3$  ということになる。群論の言葉でいえば、 $\hat{T}_{\mathbf{R}}$  を元とする巡回群の位数が有限の数  $N = N_1N_2N_3$  となる。この (15) 式と、(13) から導かれる

$$\hat{T}_{N_i\mathbf{a}_i}\psi(\mathbf{r}) = (\hat{T}_{\mathbf{a}_i})^{N_i}\psi(\mathbf{r}) = (C_{\mathbf{a}_i})^{N_i}\psi(\mathbf{r}) \quad (16)$$

を比較すると、 $C_{\mathbf{a}_i}$  は整数  $m_i$  を用いて

$$C_{\mathbf{a}_i} = \exp\left(\frac{2\pi i}{N_i}m_i\right) \quad (17)$$

と表せる。ただし  $m_i$  は  $1 \leq m_i \leq N_i - 1$  を満たすものとする。

よって、並進操作  $\hat{T}_{\mathbf{R}}$  ( $\mathbf{R} = n_1\mathbf{a}_1 + n_2\mathbf{a}_2 + n_3\mathbf{a}_3$ ) により、波動関数は

$$\hat{T}_{\mathbf{R}}\psi(\mathbf{r}) = (\hat{T}_{\mathbf{a}_1})^{n_1}(\hat{T}_{\mathbf{a}_2})^{n_2}(\hat{T}_{\mathbf{a}_3})^{n_3}\psi(\mathbf{r}) = \exp\left\{2\pi i\left(\frac{n_1m_1}{N_1} + \frac{n_2m_2}{N_2} + \frac{n_3m_3}{N_3}\right)\right\}\psi(\mathbf{r}) \quad (18)$$

のような変換を受ける。ここで、既約表現を区別するラベルとして

$$\mathbf{k}_m = \frac{m_1}{N_1}\mathbf{b}_1 + \frac{m_2}{N_2}\mathbf{b}_2 + \frac{m_3}{N_3}\mathbf{b}_3 \quad (19)$$

を導入する。 $\mathbf{b}_i$  は逆格子空間における基本ベクトルである。 $\mathbf{k}_m$  は、逆格子空間におけるブリルアンゾーン  $\mathbf{b}_1 \cdot (\mathbf{b}_2 \times \mathbf{b}_3)$  を各方向  $\mathbf{b}_i$  に沿って  $N_i$  分割したときのメッシュの点を表している。

逆格子ベクトルと実空間の並進ベクトルの関係  $\mathbf{a}_i \cdot \mathbf{b}_j = 2\pi\delta_{ij}$  を踏まえると

$$\hat{T}_{\mathbf{R}}\psi(\mathbf{r}) = \exp(i\mathbf{k}_m \cdot \mathbf{R})\psi(\mathbf{r}) \quad (20)$$

と書ける。 $N_i$  はマクロな数であり、従って  $\mathbf{k}_m$  は厳密には離散的な値を取るが、実際にはほぼ連続した値をとるとみなすことができる。そこで  $\mathbf{k}_m$  を添字を省略して  $\mathbf{k}$  と書くことにしよう。

(20) の関係式と並進操作の定義から

$$\psi(\mathbf{r} + \mathbf{R}) = \exp(i\mathbf{k} \cdot \mathbf{R})\psi(\mathbf{r}) \quad (21)$$

が成り立つことが示される。これはブロッホの定理と呼ばれる。 $\psi(\mathbf{r})$  はブロッホ関数と呼ばれる。

ブロッホ関数の形としては有名な表式がもう一つある。今、ブロッホ関数を用いて、ある関数

$$u_{\mathbf{k}}(\mathbf{r}) \equiv e^{-i\mathbf{k} \cdot \mathbf{r}}\psi(\mathbf{r}) \quad (22)$$

を定義しよう。すると、これは

$$u_{\mathbf{k}}(\mathbf{r} + \mathbf{R}) = e^{-i\mathbf{k}\cdot(\mathbf{r}+\mathbf{R})}\psi(\mathbf{r} + \mathbf{R}) = e^{-i\mathbf{k}\cdot(\mathbf{r}+\mathbf{R})}e^{i\mathbf{k}\cdot\mathbf{R}}\psi(\mathbf{r}) = u_{\mathbf{k}}(\mathbf{r}) \quad (23)$$

となるので、格子の周期性を持った関数であることが容易にわかる。この  $u_{\mathbf{k}}(\mathbf{r})$  を用いて

$$\psi(\mathbf{r}) = e^{i\mathbf{k}\cdot\mathbf{r}}u_{\mathbf{k}}(\mathbf{r}) \quad (24)$$

と表した表式もしばしばみられる。

■ **平面波を基底とした Kohn-Sham 方程式** KS 方程式を解く際、波動関数と電荷密度を平面波で展開して逆空間の問題に置き換えるのが、平面波第一原理電子状態計算である。

今、固体のように原子が周期的に並んだ系での電子の固有状態を求めたい。すなわち、周期  $\mathbf{R}$  を持つポテンシャル  $V(\mathbf{r})$  の下での一電子シュレーディンガー方程式を解くことを考える：

$$\hat{H}\psi_i(\mathbf{r}) = \epsilon_i\psi_i(\mathbf{r}), \quad \hat{H} = -\frac{1}{2}\nabla^2 + V(\mathbf{r}) \quad (25)$$

(25) の解として、波動関数を平面波  $e^{i\mathbf{k}\cdot\mathbf{r}}$  で展開した関数系を考える。

$$\psi_i(\mathbf{r}) = \sum_{\mathbf{k}} C_{\mathbf{k}}^i e^{i\mathbf{k}\cdot\mathbf{r}} \quad (26)$$

ここで、 $C_{\mathbf{k}}^i$  は展開係数である。また、ポテンシャルは格子点の周期性を持つ。格子の周期性を持った平面波の波数はまさに逆格子ベクトル  $\mathbf{G}$  である。ゆえに、ポテンシャルは

$$V(\mathbf{r}) = \sum_{\mathbf{G}} V_{\mathbf{G}} e^{i\mathbf{G}\cdot\mathbf{r}} \quad (27)$$

のように級数展開できるであろう。(26) と (27) を (25) に代入すると

$$\sum_{\mathbf{k}} e^{i\mathbf{k}\cdot\mathbf{r}} \left\{ \left( \frac{1}{2}|\mathbf{k}|^2 - \epsilon_i \right) C_{\mathbf{k}}^i + \sum_{\mathbf{G}} V_{\mathbf{G}} C_{\mathbf{k}-\mathbf{G}}^i \right\} = 0 \quad (28)$$

が得られる。よって、全ての  $\mathbf{k}$  に対して

$$\left( \frac{1}{2}|\mathbf{k}|^2 - \epsilon_i \right) C_{\mathbf{k}}^i + \sum_{\mathbf{G}} V_{\mathbf{G}} C_{\mathbf{k}-\mathbf{G}}^i = 0 \quad (29)$$

が成り立つ。(29) は、(26) の展開係数について、 $C_{\mathbf{k}}^i$  と逆格子ベクトル  $\mathbf{G}$  だけ異なったものだけを解として許していると見ることができる。これより、波動関数の展開は

$$\psi_{i\mathbf{k}}(\mathbf{r}) = \frac{1}{\sqrt{\Omega}} \sum_{\mathbf{G}} C_{\mathbf{k}-\mathbf{G}}^i \exp[i(\mathbf{k} - \mathbf{G}) \cdot \mathbf{r}] \quad (30)$$

$\mathbf{G}$  に関する和は無限個に渡るが、実際の計算においては平面波の運動エネルギー  $|\mathbf{k} + \mathbf{G}|^2/2$  があるエネルギー  $E_{cut}$  以下のものについてのみ和がとられる。平面波で展開した  $|\psi_{i\mathbf{k}}\rangle$  を Kohn-Sham 方程式に代入し、左から  $\langle \mathbf{k} + \mathbf{G} |$  を掛けると、以下の行列方程式が得られる：

$$\sum_{\mathbf{G}'} H_{\mathbf{k}+\mathbf{G}, \mathbf{k}+\mathbf{G}'} C_{\mathbf{k}+\mathbf{G}'}^i = \epsilon_{i\mathbf{k}} C_{\mathbf{k}+\mathbf{G}}^i \quad (31)$$

ここで、 $H_{\mathbf{k}+\mathbf{G}, \mathbf{k}+\mathbf{G}'} = \langle \mathbf{k} + \mathbf{G} | \hat{H} | \mathbf{k} + \mathbf{G}' \rangle$  とおいた。

全空間の体積  $\Omega$  が、体積が  $\Omega_{\text{cell}}$  のユニットセル  $N_{\text{cell}}$  個の集まり（つまり  $\Omega = N_{\text{cell}}\Omega_{\text{cell}}$ ）でであるとす  
る。(30) より、波動関数は

$$\begin{aligned}\psi_{i\mathbf{k}}(\mathbf{r}) &\equiv \langle \mathbf{r} | \psi_{i\mathbf{k}} \rangle = \frac{1}{\sqrt{\Omega}} \sum_{\mathbf{G}} C_{\mathbf{k}+\mathbf{G}}^i \exp[i(\mathbf{k} + \mathbf{G}) \cdot \mathbf{r}] \\ &= \exp(i\mathbf{k} \cdot \mathbf{r}) \frac{1}{\sqrt{N_{\text{cell}}\Omega_{\text{cell}}}} \sum_{\mathbf{G}} C_{\mathbf{k}+\mathbf{G}}^i \exp(i\mathbf{G} \cdot \mathbf{r}) \\ &\equiv \exp(i\mathbf{k} \cdot \mathbf{r}) \frac{1}{\sqrt{N_{\text{cell}}}} u_{i\mathbf{k}}(\mathbf{r})\end{aligned}\quad (32)$$

これはブロッホの定理である。

## 4 対角化アルゴリズム

第一原理電子状態計算では、Kohn-Sham (KS) 方程式と呼ばれる二階偏微分方程式を解く。QE においてこのプロセスが記述されているのが `c_bands.f90` である。

### ■ Davidson 法によるハミルトニアンに対角化

各  $\mathbf{k}$  に対し、次の固有値問題を解くことを考える：

$$\hat{H} |\psi_i\rangle = \epsilon_i \hat{S} |\psi_i\rangle, \quad i = 1, \dots, N \quad (33)$$

$\hat{H}$  は  $N \times N$  のハミルトニアン行列、 $\hat{S}$  は  $N \times N$  の重なり積分 (overlap matrix) の行列を表し、その行列要素は

$$\langle \psi_i | \hat{S} | \psi_j \rangle = \delta_{ij} \quad (34)$$

で与えられる。

$N$  が大きくなると対角化計算が容易ではなくなってくる。今、 $nbnd$  個の固有状態を得たいとする。もし、 $nbnd \ll N$  であるとする、以下で述べる反復法が有効になる。この方法では、ある行列  $A$  をベクトル  $\mathbf{b}^{(0)}$  に繰り返し作用させ

$$\mathcal{K} = \left\{ \mathbf{b}^{(0)}, A\mathbf{b}^{(0)}, \dots, A^{n-1}\mathbf{b}^{(0)} \right\} \quad (35)$$

という空間 (Krylov 部分空間) を作り、この空間における基底を用いて固有値問題を解く。この方法により、求めたい  $nbnd$  個の解を、元の基底の次元  $N$  よりも少ない数の基底を使って高精度に計算することが可能になる。

QE では、反復法の一つである davidson 法と呼ばれる方法がデフォルトとして採用されている。以下、davidson 法のアルゴリズムを示す。まず、試行ベクトル  $|\psi_i^{(0)}\rangle$  により張られる空間

$$\mathcal{K} = \left\{ |\psi_1^{(0)}\rangle, |\psi_2^{(0)}\rangle, \dots, |\psi_{nbnd}^{(0)}\rangle \right\} \quad (36)$$

と近似的な固有値  $\epsilon_i^{(0)} = \langle \psi_i^{(0)} | H | \psi_i^{(0)} \rangle$  が予め決められているとする。これを元に、次の修正ベクトル (correction vector) と呼ばれるベクトル

$$|\delta\psi_i^{(0)}\rangle = D^{-1}(H - \epsilon_i^{(n)}S) |\psi_i^{(0)}\rangle \quad (37)$$

を導入する。ここで、 $D$  は

$$D = H_{mm} \delta_{mm'} - \epsilon_i^{(n)} I \quad (38)$$

で与えられる。<sup>2</sup> 修正ベクトル  $|\delta\psi_i^{(0)}\rangle$  から、 $|\tilde{\psi}_i^{(1)}\rangle = |\psi_i^{(0)}\rangle$ ,  $|\tilde{\psi}_{i+nbnd}^{(1)}\rangle = |\delta\psi_i^{(0)}\rangle$  として、 $2nbnd$  次元の空間

$$\mathcal{K} = \left\{ |\tilde{\psi}_1^{(1)}\rangle, |\tilde{\psi}_2^{(1)}\rangle, \dots, |\tilde{\psi}_{nbnd}^{(1)}\rangle, |\tilde{\psi}_{nbnd+1}^{(1)}\rangle, |\tilde{\psi}_{nbnd+2}^{(1)}\rangle, \dots, |\tilde{\psi}_{2nbnd}^{(1)}\rangle \right\} \quad (40)$$

を作る。この拡張された空間において、ハミルトニアン行列  $\tilde{H}_{ij}^{(1)} = \langle \tilde{\psi}_i^{(1)} | H | \tilde{\psi}_j^{(1)} \rangle$  を構築し、対角化により  $nbnd$  個の固有値と固有ベクトル  $|\psi_i^{(1)}\rangle$  を求める。

こうして求めた固有ベクトルをもとに再び修正ベクトルを作り、部分空間 (40) に新たな基底として加え、ハミルトニアン行列 ( $3nbnd \times 3nbnd$ ) を作り、対角化により固有値と固有ベクトルを求める。これを、固有値が収束するまで繰り返す。

---

2

$$|R_i^{(n)}\rangle = (H - \epsilon_i^{(n)} S) |\psi_i^{(n)}\rangle \quad (39)$$

は残差ベクトル (residual vector) と呼ばれる。

## 5 対角化コードを覗く

### 5.1 c.bands.f90 の構造

サブルーチン `c_bands()` ではハミルトニアン<sup>3</sup>の対角化を行い、エネルギー固有値と Kohn-Sham 軌道を計算する。`c_bands.f90` 内で、`diag_bands_gamma` と `diag_bands_k` 内で呼び出しているサブルーチン `regterg()` と `cegterg()`<sup>3</sup> に対角化のプロセスが記述されている。これら2つのサブルーチンは `q-e-qe-5.1.0/PW/src/` 内の `regterg.f90` と `cegterg.f90` に含まれている。最後の `c_bands_nscf()` で non-scf 計算を記述している。

```
SUBROUTINE c_bands()
```

```
  k_loop: DO ik = 1, nks
    CALL diag_bands()
  END
```

```
SUBROUTINE diag_bands()
```

```
  CALL diag_bands_gamma()
```

```
    ⋮
```

```
  CALL diag_bands_k()
```

```
    ⋮
```

```
CONTAINS
```

```
  SUBROUTINE diag_bands_gamma()
```

```
    david_loop: DO
      CALL regterg()
    END david_loop
```

```
    ⋮
```

```
  SUBROUTINE diag_bands_k()
```

```
    david_loop: DO
      CALL cegterg()
    END david_loop
```

```
    ⋮
```

```
  SUBROUTINE test_exit_cond()
```

```
SUBROUTINE c_bands_efield()
```

```
SUBROUTINE c_bands_nscf()
```

<sup>3</sup>regterg = real eigen iterative generalized  
cegterg = complex eigen iterative generalized



## 5.2 cegterg.f90

davidson法アルゴリズムはcegterg.f90に記述されている。cegterg.f90は、`q-e-qe-5.1.0/PW/src/`内にある。

### ■ 変数 (12~92 行目)

```
12 !-----
13 SUBROUTINE cegterg( npw, npwx, nvec, nvecx, npol, evc, ethr, &
14                   uspp, e, btype, notcnv, lrot, dav_iter )
15 !-----
16 !
17 ! ... iterative solution of the eigenvalue problem:
18 !
19 ! ... ( H - e S ) * evc = 0
20 !
21 ! ... where H is an hermitean operator, e is a real scalar,
22 ! ... S is an overlap matrix, evc is a complex vector
23 !
24 USE kinds, ONLY : DP
25 USE mp_bands , ONLY : intra_bgrp_comm
26 USE mp, ONLY : mp_sum
27 !
28 IMPLICIT NONE
29 !
30 INTEGER, INTENT(IN) :: npw, npwx, nvec, nvecx, npol
31     ! dimension of the matrix to be diagonalized
32     ! leading dimension of matrix evc, as declared in the calling pgm unit
33     ! integer number of searched low-lying roots
34     ! maximum dimension of the reduced basis set :
35     ! (the basis set is refreshed when its dimension would exceed nvecx)
36     ! umber of spin polarizations
37 COMPLEX(DP), INTENT(INOUT) :: evc(npwx,npol,nvec)
38     ! evc contains the refined estimates of the eigenvectors
39 REAL(DP), INTENT(IN) :: ethr
40     ! energy threshold for convergence :
41     ! root improvement is stopped, when two consecutive estimates of the root
42     ! differ by less than ethr.
43 LOGICAL, INTENT(IN) :: uspp
44     ! if .FALSE. : do not calculate S|psi>
45 INTEGER, INTENT(IN) :: btype(nvec)
46     ! band type ( 1 = occupied, 0 = empty )
47 LOGICAL, INTENT(IN) :: lrot
48     ! .TRUE. if the wfc have already been rotated
49 REAL(DP), INTENT(OUT) :: e(nvec)
50     ! contains the estimated roots.
51 INTEGER, INTENT(OUT) :: dav_iter, notcnv
52     ! integer number of iterations performed
53     ! number of unconverged roots
54 !
55 ! ... LOCAL variables
56 !
57 INTEGER, PARAMETER :: maxter = 20
58     ! maximum number of iterations
```

```

59  !
60  INTEGER :: kterm, nbase, np, kdim, kdmx, n, m, nb1, nbn
61  ! counter on iterations
62  ! dimension of the reduced basis
63  ! counter on the reduced basis vectors
64  ! adapted npw and npwx
65  ! do-loop counters
66  INTEGER :: ierr
67  COMPLEX(DP), ALLOCATABLE :: hc(:,,:), sc(:,,:), vc(:,,:)
68  ! Hamiltonian on the reduced basis
69  ! S matrix on the reduced basis
70  ! the eigenvectors of the Hamiltonian
71  COMPLEX(DP), ALLOCATABLE :: psi(:,,:), hpsi(:,,:), spsi(:,,:)
72  ! work space, contains psi
73  ! the product of H and psi
74  ! the product of S and psi
75  REAL(DP), ALLOCATABLE :: ew(:)
76  ! eigenvalues of the reduced hamiltonian
77  LOGICAL, ALLOCATABLE :: conv(:)
78  ! true if the root is converged
79  REAL(DP) :: empty_ether
80  ! threshold for empty bands
81  !
82  REAL(DP), EXTERNAL :: ddot
83  !
84  ! EXTERNAL h_psi, s_psi, g_psi
85  ! h_psi(npwx,npw,nvec,psi,hpsi)
86  ! calculates H|psi>
87  ! s_psi(npwx,npw,nvec,spsi)
88  ! calculates S|psi> (if needed)
89  ! Vectors psi,hpsi,spsi are dimensioned (npwx,npol,nvec)
90  ! g_psi(npwx,npw,notcnv,psi,e)
91  ! calculates (diag(h)-e)^-1 * psi, diagonal approx. to (h-e)^-1*psi
92  ! the first nvec columns contain the trial eigenvectors

```

変数	意味
npw	ある $k$ における平面波の数
npwx	ある $k$ における平面波の数の最大数
nvec	求める固有状態の数。c_band() の nbnd から引き継がれる。
nvecx	
npol	
evc(npwx,npol,nvec)	
uspp	論理変数。ウルトラソフト擬ポテンシャル法の場合、真
btype(nvec)	
lrot	
e(nvec)	バンド nvec における固有値
dav_iter	
notcnv	
nbase	

## ■ 初期の固有値と固有ベクトルの推定 (12~92 行目)

```

143 notcnv = nvec
144 nbase = nvec
145 conv = .FALSE.
146 !
147 IF ( uspp ) spsi = ZERO
148 !
149 hpsi = ZERO
150 psi = ZERO
151 psi(:, :, 1:nvec) = evc(:, :, 1:nvec)
152 !
153 ! ... hpsi contains h times the basis vectors
154 !
155 CALL h_psi( npwx, npw, nvec, psi, hpsi )
156 !
157 ! ... spsi contains s times the basis vectors
158 !
159 IF ( uspp ) CALL s_psi( npwx, npw, nvec, psi, spsi )

```

151 行目で、何らかの方法で推定された初期基底  $evc$  を、配列  $psi$  に代入する。

$$\mathcal{K} = \left\{ \left| \psi_1^{(0)} \right\rangle, \left| \psi_2^{(0)} \right\rangle, \dots, \left| \psi_{nvec}^{(0)} \right\rangle \right\} \quad (41)$$

$\left| \psi_i^{(0)} \right\rangle$  は列ベクトルで表記すると

$$\left| \psi_i^{(0)} \right\rangle \rightarrow \begin{bmatrix} \left\langle \mathbf{k} + \mathbf{G}_1 \left| \psi_i^{(0)} \right\rangle \right. \\ \left\langle \mathbf{k} + \mathbf{G}_2 \left| \psi_i^{(0)} \right\rangle \right. \\ \vdots \\ \left\langle \mathbf{k} + \mathbf{G}_{npw} \left| \psi_i^{(0)} \right\rangle \right. \end{bmatrix} \quad (42)$$

のように表される。この  $psi$  に対し、155 行目でコールされるサブルーチン  $hpsi$  ( $spsi$ ) で  $H(S)$  と基底ベクトル  $psi$  の積が出力される。

$$\left\{ H \left| \psi_1^{(0)} \right\rangle, H \left| \psi_2^{(0)} \right\rangle, \dots, H \left| \psi_{nvec}^{(0)} \right\rangle \right\} \quad (43)$$

なお、 $hpsi$  ( $spsi$ ) は `q-e-qe-5.1.0/PW/src/` 中にある `h.psi.f90` (`s.psi.f90`) に記述されている。

```

160 !
161 ! ... hc contains the projection of the hamiltonian onto the reduced
162 ! ... space vc contains the eigenvectors of hc
163 !
164 hc(:, :) = ZERO
165 sc(:, :) = ZERO
166 vc(:, :) = ZERO
167 !
168 CALL ZGEMM( 'C', 'N', nbase, nbase, kdim, ONE, &
169           psi, kdmx, hpsi, kdmx, ZERO, hc, nvecx )
170 !
171 CALL mp_sum( hc( :, 1:nbase ), intra_bgrp_comm )
172 !

```

```

173  IF ( uspp ) THEN
174      !
175      CALL ZGEMM( 'C', 'N', nbase, nbase, kdim, ONE, &
176                psi, kdmx, spsi, kdmx, ZERO, sc, nvecx )
177      !
178  ELSE
179      !
180      CALL ZGEMM( 'C', 'N', nbase, nbase, kdim, ONE, &
181                psi, kdmx, psi, kdmx, ZERO, sc, nvecx )
182      !
183  END IF
184      !
185  CALL mp_sum( sc( :, 1:nbase ), intra_bgrp_comm )
186      !
187  IF ( lrot ) THEN
188      !
189      DO n = 1, nbase
190          !
191          e(n) = REAL( hc(n,n) )
192          !
193          vc(n,n) = ONE
194          !
195      END DO
196      !
197  ELSE
198      !
199      ! ... diagonalize the reduced hamiltonian
200      !
201      CALL cdiaghg( nbase, nvec, hc, sc, nvecx, ew, vc )
202      !
203      e(1:nvec) = ew(1:nvec)
204      !
205  END IF

```

168、175、180 行目で呼び出される ZGEMM において、hpsi と spsi に代入された  $H|\psi_i^{(0)}\rangle$  と  $S|\psi_i^{(0)}\rangle$  にブラ  $\langle\psi_i^{(0)}|$  を掛けることでハミルトニアンと重なり積分の行列要素が計算され、配列 hc と sc に代入される。ZGEMM というのは、LAPACK(Linear Algebra PACKage) と呼ばれる線形代数の演算のパッケージにおけるプログラムの一つであり、複素行列の積を計算する。<sup>4</sup> ここでは、行列 (43) と  $|\psi_i^{(0)}\rangle$  に共役なベクトルを入力として、以下の行列

$$\begin{bmatrix} \langle\psi_1^{(0)}|H|\psi_1^{(0)}\rangle & \cdots & \langle\psi_1^{(0)}|H|\psi_{nbase}^{(0)}\rangle \\ \vdots & & \vdots \\ \langle\psi_{nbase}^{(0)}|H|\psi_1^{(0)}\rangle & \cdots & \langle\psi_{nbase}^{(0)}|H|\psi_{nbase}^{(0)}\rangle \end{bmatrix} \quad (44)$$

を返す (S の方も同様)。

189～205 行目で初期の近似的な固有値と固有ベクトルが決定される。lrot が真であれば、試行ベクトル  $|\psi_i^{(0)}\rangle$  により

$$\epsilon_i^{(0)} = \langle\psi_i^{(0)}|H|\psi_i^{(0)}\rangle \quad (45)$$

<sup>4</sup>具体的な計算内容は、<https://netlib.org/lapack/explore-html-3.6.1/dc/d17/> のサイトから、Modules → Level3 complex16 にアクセスして確認してほしい。

が固有値として配列 e(n) に代入され、固有ベクトルは単位行列として配列 vc に代入される。lrot が偽であれば、201 行目でコールされる cdiaghg にて行列 (44) を対角化し、得られた固有値 ew が e(n) に代入される。

#### ■ 基底の拡張 (206~299 行目)

---

```

206  !
207  ! ... iterate
208  !
209  iterate: DO kter = 1, maxter
210      !
211      dav_iter = kter
212      !
213      CALL start_clock( 'cegterg:update' )
214      !
215      np = 0
216      !
217      DO n = 1, nvec
218          !
219          IF ( .NOT. conv(n) ) THEN
220              !
221              ! ... this root not yet converged ...
222              !
223              np = np + 1
224              !
225              ! ... reorder eigenvectors so that coefficients for unconverged
226              ! ... roots come first. This allows to use quick matrix-matrix
227              ! ... multiplications to set a new basis vector (see below)
228              !
229              IF ( np /= n ) vc(:,np) = vc(:,n)
230              !
231              ! ... for use in g_psi
232              !
233              ew(nbase+np) = e(n)
234              !
235          END IF
236          !
237      END DO

```

---

```

238      !
239      nb1 = nbase + 1
240      !
241      ! ... expand the basis set with new basis vectors ( H - e*S )|psi> ...
242      !
243      IF ( uspp ) THEN
244          !
245          CALL ZGEMM( 'N', 'N', kdim, notcnv, nbase, ONE, spsi, &
246                   kdmx, vc, nvecx, ZERO, psi(1,1,nb1), kdmx )
247          !
248      ELSE
249          !
250          CALL ZGEMM( 'N', 'N', kdim, notcnv, nbase, ONE, psi, &
251                   kdmx, vc, nvecx, ZERO, psi(1,1,nb1), kdmx )

```

```

252      !
253      END IF
254      !
255      DO np = 1, notcnv
256          !
257          psi(:, :, nbase+np) = - ew(nbase+np)*psi(:, :, nbase+np)
258          !
259      END DO
260      !
261      CALL ZGEMM( 'N', 'N', kdim, notcnv, nbase, ONE, hpsi, &
262                kdmx, vc, nvecx, ONE, psi(1,1,nb1), kdmx )
263      !
264      CALL stop_clock( 'cegterg:update' )

```

$kter$  回目のイタレーションにおいて、一つ前までのイタレーションで、notcnv 個の固有ベクトル  $|\psi_i^{(kter-1)}\rangle$  が配列 vc に格納されている。これらの固有ベクトルに行列  $(H - eS)$  を掛けることで notcnv 個の修正ベクトルを作る。

245 行目で  $S|\psi_i\rangle$  が計算され、257 行目でこれと近似固有値の積  $-eS|\psi_i\rangle$  が計算される。この結果は配列 psi(1,1,nb1) に格納される。261 行目の ZGEMM では  $-e_i S|\psi_i\rangle$  に  $H|\psi_i\rangle$  を加えた結果が返される。

ここまでで、

$$(H - e_i S) |\psi_i^{(kter-1)}\rangle \quad (i = nbase + 1, \dots, nbase + notcnv) \quad (46)$$

が得られたことになる。

```

265      !
266      ! ... approximate inverse iteration
267      !
268      CALL g_psi( npwx, npw, notcnv, npol, psi(1,1,nb1), ew(nb1) )
269      !

```

(46) に行列  $D$  の逆行列を掛け、修正ベクトルを作る。

$$|\delta\psi_i^{(kter-1)}\rangle = D^{-1}(H - \epsilon_i S) |\psi_i^{(kter-1)}\rangle \quad (47)$$

```

270      ! ... "normalize" correction vectors psi(:,nb1:nbase+notcnv) in
271      ! ... order to improve numerical stability of subspace diagonalization
272      ! ... (cdiaghg) ew is used as work array :
273      !
274      ! ... ew = <psi_i|psi_i>, i = nbase + 1, nbase + notcnv
275      !
276      DO n = 1, notcnv
277          !
278          nbn = nbase + n
279          !
280          IF ( npol == 1 ) THEN
281              !
282              ew(n) = ddot( 2*npw, psi(1,1,nbn), 1, psi(1,1,nbn), 1 )
283              !
284          ELSE
285              !
286              ew(n) = ddot( 2*npw, psi(1,1,nbn), 1, psi(1,1,nbn), 1 ) + &
287                      ddot( 2*npw, psi(1,2,nbn), 1, psi(1,2,nbn), 1 )

```

```

288         !
289         END IF
290         !
291     END DO
292     !
293     CALL mp_sum( ew( 1:notcnv ), intra_bgrp_comm )
294     !
295     DO n = 1, notcnv
296         !
297         psi(:, :, nbase+n) = psi(:, :, nbase+n) / SQRT( ew(n) )
298         !
299     END DO

```

280～297 行目では、新たに追加されたベクトルの規格化が行われている。以上より、

$$|\tilde{\psi}_i^{(kter)}\rangle = |\tilde{\psi}_i^{(kter-1)}\rangle, \quad |\tilde{\psi}_{i+notcnv}^{(kter)}\rangle = |\delta\psi_i^{(kter-1)}\rangle \quad (48)$$

とし、部分空間を拡張して

$$\mathcal{K} = \left\{ |\tilde{\psi}_1^{(kter)}\rangle, \dots, |\tilde{\psi}_{nbase}^{(kter)}\rangle, |\tilde{\psi}_{nbase+1}^{(kter)}\rangle, \dots, |\tilde{\psi}_{nbase+notcnv}^{(kter)}\rangle \right\} \quad (49)$$

を得る。

#### ■ 拡張された空間における行列の構成と対角化 (300～376 行目)

```

300     !
301     ! ... here compute the hpsi and spsi of the new functions
302     !
303     !
304     CALL h_psi( npwx, npw, notcnv, psi(1,1,nb1), hpsi(1,1,nb1) )
305     !
306     IF ( uspp ) &
307         CALL s_psi( npwx, npw, notcnv, psi(1,1,nb1), spsi(1,1,nb1) )
308     !
309     ! ... update the reduced hamiltonian
310     !
311     CALL start_clock( 'cegterg:overlap' )
312     !
313     CALL ZGEMM( 'C', 'N', nbase+notcnv, notcnv, kdim, ONE, psi, &
314         kdmx, hpsi(1,1,nb1), kdmx, ZERO, hc(1,nb1), nvecx )
315     !
316     CALL mp_sum( hc( :, nb1:nb1+notcnv-1 ), intra_bgrp_comm )
317     !
318     IF ( uspp ) THEN
319         !
320         CALL ZGEMM( 'C', 'N', nbase+notcnv, notcnv, kdim, ONE, psi, &
321             kdmx, spsi(1,1,nb1), kdmx, ZERO, sc(1,nb1), nvecx )
322         !
323     ELSE
324         !
325         CALL ZGEMM( 'C', 'N', nbase+notcnv, notcnv, kdim, ONE, psi, &
326             kdmx, psi(1,1,nb1), kdmx, ZERO, sc(1,nb1), nvecx )
327         !
328     END IF

```

```

329      !
330      CALL mp_sum( sc( :, nb1:nb1+notcnv-1 ), intra_bgrp_comm )
331      !
332      CALL stop_clock( 'cegterg:overlap' )
333      !
334      nbase = nbase + notcnv
335      !
336      DO n = 1, nbase
337          !
338          ! ... the diagonal of hc and sc must be strictly real
339          !
340          hc(n,n) = CMPLX( REAL( hc(n,n) ), 0.DO ,kind=DP)
341          sc(n,n) = CMPLX( REAL( sc(n,n) ), 0.DO ,kind=DP)
342          !
343          DO m = n + 1, nbase
344              !
345              hc(m,n) = CONJG( hc(n,m) )
346              sc(m,n) = CONJG( sc(n,m) )
347              !
348          END DO
349          !
350      END DO
351      !
352      ! ... diagonalize the reduced hamiltonian
353      !
354      CALL cdiaghg( nbase, nvec, hc, sc, nvecx, ew, vc )
355      !
356      ! ... test for convergence
357      !
358      WHERE( btype(1:nvec) == 1 )
359          !
360          conv(1:nvec) = ( ( ABS( ew(1:nvec) - e(1:nvec) ) < ethr ) )
361          !
362      ELSEWHERE
363          !
364          conv(1:nvec) = ( ( ABS( ew(1:nvec) - e(1:nvec) ) < empty_ethr ) )
365          !
366      END WHERE
367      !
368      notcnv = COUNT( .NOT. conv(:) )
369      !
370      e(1:nvec) = ew(1:nvec)
371      !
372      ! ... if overall convergence has been achieved, or the dimension of
373      ! ... the reduced basis set is becoming too large, or in any case if
374      ! ... we are at the last iteration refresh the basis set. i.e. replace
375      ! ... the first nvec elements with the current estimate of the
376      ! ... eigenvectors; set the basis dimension to nvec.

```

304 行目で、新たに追加された基底ベクトル群

$$\left\{ \left| \tilde{\psi}_{nbase+1}^{(kter)} \right\rangle, \dots, \left| \tilde{\psi}_{nbase+notcnv}^{(kter)} \right\rangle \right\} \quad (50)$$



に対してハミルトニアンを掛ける（Sも同様）。その結果を入力として、313行目のZGEMMで行列

$$\begin{bmatrix} \langle \tilde{\psi}_1^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_1^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ \vdots & & \vdots \\ \langle \tilde{\psi}_{nbase}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase}^{(0)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ \langle \tilde{\psi}_{nbase+1}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase+1}^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ \vdots & & \vdots \\ \langle \tilde{\psi}_{nbase+notcnv}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase+notcnv}^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \end{bmatrix} \quad (51)$$

を計算し、配列 hc(1,1,nb1) に代入する。ここまでの、配列 hc は次のような行列を表すものとなっている：

$$\begin{bmatrix} \langle \tilde{\psi}_1^{(kter-1)} | H | \tilde{\psi}_1^{(kter-1)} \rangle & \cdots & \langle \tilde{\psi}_1^{(kter-1)} | H | \tilde{\psi}_{nbase}^{(kter-1)} \rangle & \langle \tilde{\psi}_1^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_1^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ \vdots & & \vdots & \vdots & & \vdots \\ \langle \tilde{\psi}_{nbase}^{(kter-1)} | H | \tilde{\psi}_1^{(kter-1)} \rangle & \cdots & \langle \tilde{\psi}_{nbase}^{(kter-1)} | H | \tilde{\psi}_{nbase}^{(kter-1)} \rangle & \langle \tilde{\psi}_{nbase}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase}^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ 0 & \cdots & 0 & \langle \tilde{\psi}_{nbase+1}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase+1}^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \\ \vdots & & \vdots & & & \vdots \\ 0 & \cdots & 0 & \langle \tilde{\psi}_{nbase+notcnv}^{(kter)} | H | \tilde{\psi}_{nbase+1}^{(kter)} \rangle & \cdots & \langle \tilde{\psi}_{nbase+notcnv}^{(kter)} | H | \tilde{\psi}_{nbase+notcnv}^{(kter)} \rangle \end{bmatrix} \quad (52)$$

336～350行目では、hcの対角成分が実数になっているかチェックされる。354行目で、この拡大された行列に対し、対角化計算を行う。358行目以降は、固有値が収束しているかどうかで、収束判定がなされる。